
Overview

Although QA Partner’s 4Test language has the ability to control up to 255 separate computers for simultaneous testing, there is very little information available on putting the practice into use and, more importantly, how to convert existing single-machine automation code into code that can operate in the multiple machine environment without completely re-writing the code. Ultimately, the procedures documented below had to be ferreted out in an exploratory way by trial and error, but the end result is a surprisingly elegant and simple way to salvage single machine code with minimal editing. This paper intends to not only demonstrate the method but to lead the reader through an understanding of each element and why it is used the way it is. This article presumes that you are already familiar with coding in 4-test for single machine use and is not a beginners tutorial.
Background - Segue QA Partner behavior

Standalone machine testing concepts.

The Agent

In the standard installation of QA Partner, an “agent” is installed on each machine. This agent communicates directly to the operating system of that machine and is specific to that operating system. There are agents for Windows, Macintosh, Unix, etc. Separate from the agent is QA Partner itself (or QA Partner plus Organizer) which not only creates, edits and complies scripts, it also sends the output of the complied version of the script to the agent or agents. This output is recognizable on any platform and it is the job of the agent to translate it into commands that are specific to the local operating system. This makes it possible for any type of supported machine (windows. Mac, Unix etc.) to run a script that can tell any other kind of supported machine what to do. One control (or host) machine can send instructions to up to 255 target machines on any mix of supported platforms. Each target machine needs to have an agent (and there is a separate license for each), but only one master copy of QA Partner is needed to run them all.

The agent is automatically installed when QA partner is installed. Separately purchased agents are installed separately. One confusing surprise might be that on first startup, QA partner doesn’t seem to work. It is usually the case that the installation does not create an automatic startup for the agent although, you must have both QAP and the agent running to successfully run a script. Therefore you have to start both separately, or preferably, add the agent to the startup folder and always have it present.

The Frame

In the standard usage of QA Partner, a basic single-machine script assumes that a “frame’ has been recorded or created. This is an include file (frame.inc) that holds declarations of all the objects in the application to be tested. QA Partner can batch-create the first level of objects (main windows and menu items) but you must individually record, and perhaps manually tweak, the deeper dialog boxes and menu items. For instance, you might add a procedure to automatically handle security logon dialogs at startup. The frame can also refer to other include files as needed.
Testcases

Segue expects that tests are written using special procedures of the type ‘testcase’. The tests themselves are written in the form of procedures that are called by the main script. As in any programming languages, procedures can call on other procedures and pass variables. Thus building blocks can be created to develop more complex procedures. In addition to normal programming concepts, testcase procedures have additional built-in features specific to testing.

BaseState

The base state is a snapshot definition of the application to be tested when it is first started and ready to use. This definition is included in the frame. When a testcase is invoked, the frame is automatically checked for a declaration of the ‘base state’.  If the application is not in the specified base state, the application is started and/or is navigated to that point.  This is used to assure that each test starts at a known point and that a given test is not invalidated by unexpected results in a previous test. Having a defined base state is essential for proper error recovery.
AppState

One can also define ‘appstates’ which are common starting points for many tests that are deeper into the application (e.g. at the file save dialog). A given testcase can be told to begin at any of a list of user-defined appstates, saving time and trouble.

Using these starting states, the testcase then navigates to the point of test, applies the test, tries to verify the result of the test using one or more of many verify commands, and finally navigates back to the base state. If an error is detected by any of the verify functions, the testcase ends right there, an error message is written to the results file, the internal testcase error recovery functions return the application to the base state, and the next testcase is invoked. If no error is found, typically nothing is written to the results file (depending on user settings) and the next test proceeds. You can add commands to write notes and test results to the results file, however you can also set runtime options to record fine detail information to the results file on each instruction to the agent(s).

Moving on to multi-machine testing.

Every machine to be used needs its own installation of a QA Partner agent in the version appropriate to the operating system of that machine. This document presumes the use of Windows NT 4.0. The target machines only need the agent and not the full QA partner of the control machine. For multi-machine testing, the agents need an additional configuration step after installation.

Agent configuration

Start the agent if it is not already running. Open the agent and you will see a featureless box. Right mouse the upper left icon to drop down a menu, and select network. Change Network from (none) to TCP/IP.  The port number 2965 should also appear. Close the box. As part of any multi-machine testing setup, each machine’s network name is attached to its agent, giving each agent/machine combination a unique network name. In this example the machines are moe, larry, and curly and the agent names are the same. 

Frames

If one application is to be tested on many machines, one frame is sufficient for all (presuming that the executable files are located in the same path on each machine - e.g. each machine has myapp.exe located in c:\mytestdirectory\). When multiple applications are to be tested then one frame for each kind of application must be recorded/created. I have not tested a case where the path to the executable differs among the target machines.

The basic scenario documented in the QAP User Guide presumes, without saying so, that a fixed number of machines are being used, therefore programming the relationship can be hard coded. If your list of available machines is variable, you have to handle this in a more elaborate way. A solution for this situation is presented below.

Connection and multitestcase()

The Segue documentation does not directly point out the role of connection. In single-machine testing the concept of connection does not come up, as you are always connected to the machine you are testing on.  In multi-machine testing the connection to each target machine must be established before each test, then broken after each test. 
Under Options|Runtime your agent is set to (local). In single machine testing, your tests use a function of the type ‘testcase()’ which calls into play the recovery system, and assures that the application is in the base state (or app state) before each test begins. However, in multiple machine testing, instead of testcase(), you use multitestcase(). At the beginning of each multitestcase(), your code must first connect to each machine and establish what the base state is for each machine. Then perform your test, verify, and clean up after your test by returning to the base state for each machine. Finally disconnect from each machine. This connect-test-disconnect procedure repeats for each and every test - perhaps thousands of times.

The connect-setup statement SetUpMachine()  connects to a machine and establishes a base state for that machine, independent of any other machine. You need one SetUpMachine() statement for each machine. They can be used to run one application on all machines, or run a different application on each machine - each with it’s unique base state based on it’s unique frame, or any mixture of instances of applications.  The code keeps them straight based on the Main window name, but things get complicated when there is more than one main window open at once.

A frame for Text Editor, for example, declares the main window name as TextEditor.  Any number of instances of this application can use the same frame by referencing this name. The name Calculator would reference the frame for Calculator. This implies that there will be a conflict if two different applications had main windows with the same name (e.g. Welcome). I have not tested handling two concurrent but dissimilar frames with different wMainWindow definitions.

B: Multitestcase concepts

Simple form of a multitestcase.

	multitestcase sample ( )     // this declares the multitestcase

   //required setup functions

   [ ] SetUpMachine(machine1[,mainwindow[,appstate]]) //one for each machine

   [ ] SetUpMachine(machine2[,mainwindow[,appstate]])

   [ ] SetMultiAppStates ()                                             //once after all the setup statements

     // YOUR TEST STATEMENTS HERE           //use loops to repeat test on each machine

                                                                               //or address each machine individually

     // Verification statements here, if not built into test functions above

     // Steps to return app to base state

   [ ] DisconnectAll ()                                     // undoes all the setups at once

main ()

     sample ()                             // this calls the multitestcase named sample




This seems to imply that a script designed for single machine testing, using testcase, will not be quickly converted to multitestcase as EACH testcase must be converted to include this connect-disconnect scheme, as well as either manually directing each test at each machine or have a ‘for each’ loop wrapped around the actual test. But take heart! It may be that most single machine tests can be used unmodified, but we have to learn some more concepts and structures before we get there. We’ll tinker with the above multitestcase structure several times before we’re done.

Setting the local agent

Will the control machine also have tests run on itself or just on target machines? On your control machine Under Options|Runtime you can set the relationship of the local agent to the QAPartner program. The default installation sets the local agent as (local) which is appropriate when you run tests on one machine alone. For multi-machine testing, you have choices depending on what you do.

If you are multi-machine testing, but your host machine only runs the script while the tests run on just the remote “target” machines, this setting can be left on (local). 

If the host machine will also be used to run tests as well as run scripts, you have to change this setting to (none) and include the name of your host machine in the list of machine to connect to. e.g. I have larry, curly and moe. curly is the host machine (no caps in the names in my convention). On curly I set the agent to (none) and my list of machines is larry, moe and curly. I could also set the agent to curly and only have larry and moe on my list, but then I’d have to maintain one list for connection loops that excludes curly (since curly would always be connected) and another list for test execution loops that includes curly (as each machine is addressed separately..

Dealing with an unpredictable list of available machines.

Is the list or number of test machines always the same or do you have to handle a varying list?

Conditional connection is complicated by the fact that you need to be able to tell if a machine is not present before you attempt to connect to it. A script will halt if it attempts to connect to an unavailable machine. How can you test for the availability of a machine that may not be visible to you elsewhere in the building? If you’re using NT, one solution is to have present, only on available machines, a unique file such as a zero-byte file named qa_ready.flg.  You can test for the presence of this file before connecting to a machine using low-level network commands.  The QAP command SYS_FileExists() can be used to create a list of currently available machines as follows. Note: local agent is set to (none).

const HOST = “curly”

LIST OF STRING lsOnLine lsNotOnLine lsMachine

[-] lsMachine = {...}                       //seed with list of all possible machines 

     [ ] “larry”

     [ ] “moe”

     [ ] “curly”

[  ] Connect (HOST)                       //must have host agent connected for the SYS_ call below to work

STRING sCurMach

[-] for each sCurMach in lsMachine         //check every potential machine


[-]  if ("{SYS_FileExists ('\\{sCurMach}\coreqa\qa_ready.flg')}" = = "TRUE") //network call



[ ] ListAppend( lsOnLine, sCurMach )


[-]  else



[ ] ListAppend( lsNotOnLine, sCurMach )

[  ] Disconnect (HOST)

Using lsOnLine, you can connect to and send commands to each of the machines that were available.

For some reason, when you set the agent of the local machine as (local) in the Options|Runtime dialog you cannot successfully use SYS_FileExists(). This also was only tested on NT4.0 and may not work on other windows platforms.

Spawning

It is highly valuable to use ‘for loops’ and the spawn & rendezvous commands, which invoke multiple threads, to speed up processing and to make conflict testing possible. Addressing the same test to multiple machines is a little tricky and the documentation is a little too simplistic. Most of the following was discovered by trial and error.

Without spawning - testing multiple machines serially:

Here is a typical serial do-loop using “for each” to apply a hypothetical function CreateText() to each machine in turn.  Each machine executes and completes before the next machine begins. This is a slow process. This fragment assumes that connection is already established and uses SetMachine() to set focus.

	// assume lsOnLine has been populated as shown above and all connections established

STRING sCurMach

 [-] for each sCurMach in lsOnLine 


[ ] SetMachine (sCurMach)


[ ] CreateText()


Clumsy unlooped Spawn:

This is the example from the book. It works and gives the basic idea of spawn

spawn


[ ] SetMachine ("larry")


[ ] CreateText()

spawn


[ ] SetMachine ("moe")


[ ] CreateText()

spawn


[ ] SetMachine ("curly")


[ ] CreateText()

rendezvous

Each spawn statement launches a “thread” which is a separate and independent execution of code. The thread directed at larry is invoked, then the thread directed at moe is invoked without waiting for, or even being aware of, what is happening with larry.  The instruction code executes so fast, to you it will appear as if all threads execute simultaneously.  The rendezvous statement pauses before the execution of any subsequent code until all the open threads complete. The processing then continues with the statements that follow rendezvous. Under Options|Runtime you can set the reporting to the results file to include the name of the thread running for each call and see that while the basic machine operation is T0, when threads are spawned they become T1, T2, T3, etc.

Can we put this in a loop? Yes, but it is tricky (everything is tricky!)

Incorrect spawn:

Do not declare loop variables globally or outside the function.

Although the following seems to make sense, it doesn’t work correctly. Note that sCurMach and lsOnLine are declared globally at the beginning of the script.

[+] LIST OF STRING lsOnLine {...}  
// globally declares list of machines - this is OK.

[  ] STRING sCurMach  


// globally declares variable - not a good idea

testcase wrong()

   [-] for each sCurMach in lsOnLine

         spawn


[ ] SetMachine (sCurMach)


[ ] CreateText()

   rendezvous

First, what’s right. Notice that programming steps after spawn are indented just like programming steps are indented after an “if” or a “for each”.  Second notice that the rendezvous statement is out-dented to the same level as the initial for statement, one level higher than the spawn.  This is correct. Still, it won’t work properly. What will usually happen with this loop is that most or all of the programming steps will be directed at curly, the last machine listed in lsOnLine.  Why? By declaring sCurMach globally at the top of the script, there is but only one variable named sCurMach and it is in reality a memory address. When each loop is using the same external variable, the current value at any moment is unpredictable except that it most likely will have run through the for list for each thread and ended on the last entry before the subsequent steps poll this address to use it. When any thread calls on the variable, no matter which thread, it will find the current value, probably the last value.  The correct procedure is to declare the variable locally, inside the testcase. Yes, you have to re-declare it inside each testcase or multitestcase. 

[+] LIST OF STRING lsOnLine {...}  
// globally declares list of machines

testcase right()

   STRING sCurMach                                   // locally declares the variable
   [-] for each sCurMach in lsOnLine


[-] spawn



[ ] SetMachine (sCurMach)



[ ] CreateText()

rendezvous

Now each thread has its own temporary version of sCurMach and keeps track of it independently of the other threads.  Each thread is consistent.

There is also a “parallel” command that you should be able to use instead of spawn-rendezvous but after early problems I haven’t returned to test it out thoroughly yet.

Spawn - rendezvous is essential if you want to test stress and conflict resolution.  You can have each machine navigate up to a critical point in one spawn, rendezvous everyone, then broadcast the one executing stroke (enter, tab, whatever) with another spawn so that differences in machine execution time are minimized and all the active machines get the same execute command in an eyeblink. Another rendezvous and a third spawn can then wait for all machines to complete, then test for results.

Multitestcase setup.

To return to our multitestcase discussion, you have to send an individual SetUpMachine command to each available machine at the beginning of each multitestcase.  The first simplification of this is to use spawn and a loop to do this and put is all into one function that you can call repeatedly. This setup can be specific for the product you are testing or it can be generalized for all applications. For demo we will use the Text Editor that ships with QA Partner. Note that the SetUpMachine command also allows you to specify base state or appstate in the setup. Although it is optional, I recommend that you always specify a state, even if it is always base state, for future ease of modification of your code. If you construct your setup function to accept app state as an input variable, one function can be used for both base state and any app states that you might develop in the future. 

Inside the function I use the wMainWindow constant from the frame you included into the test so it automatically is speaking to the application you are testing.  This form of the function would only be used for tests that are run on the same application on all machines. A different function could leave one machine out of the list and have it always be setup for a second application (say a server program).

Here’s an example

// note lsOnLine is global, having been declared in an include file

// note sMainWindow is declared in frame.inc

use frame.inc

MultiSetup(string sState)


string sCrMach


for each sCrMach in lsOnLine



SetUpMachine (sCrMach, wMainWindow, sState) 


SetMultiAppStates()

multitestcase MyMultiTest()


MultiSetup(“DefaultBaseState”)


string sCMach


for each sCMach in lsOnLine



spawn




SetMachine (sCMach)




test() //test code here either as a call or actual steps


rendezvous


DisconnectAll( )

main ()

  MyMultiTest()

Note that I had to use different variable names for the machine name in MyMultiTest and MultiSetup. Since one is called from within the other, both variables will exist simultaneously and therefore cannot have the same name.
Eliminating redundancy

The problem with ...

multitestcase MyTest()


MultiSetup(“BaseState”)


string sCurMach


for each sCurMach in lsOnLine



spawn




SetMachine (sCurMach)




test() //test code here either as a call or actual steps


rendezvous


DisconnectAll( )

main ()

  MyMultiTest()

... is that you wind up cutting and pasting the structure for each test -  hundreds or thousands of times repeatedly, only changing the Test() line or putting the actual test code there. That’s a lot of duplication of effort.

Here is a more elegant solution

Test1()


//a test to apply to all machines

Test2()


//another test to apply to all machines

multitestcase mMultiTest(string sState1,  string testname)  
//create this once


MultiSetup(sState1)


string sCurmach


for each sCurMach in lsOnLine



spawn




SetMachine (sCurMach)




@(testname) ()  // this calls the supplied testname by reference


rendezvous


DisconnectAll()

main()


mMultiTest ( “DefaultBaseState”, "Test1")  //call generic function and feed it names & states of tests


mMultiTest ( “DefaultBaseState”, "Test2")

One structure provides the repeated lines and the name of the test to be called is called by reference with the @ operator. We do wind up with a new redundancy, albeit a simpler one.  You have to cut and paste the mMultiTest call over and over with new arguments. In a moment I’ll show you a simplification for this which will have some uses.

This structure is based on the concept that a complete test is contained in a single call, as in..

Test1() 
\\ starts from base/appstate, navigates to test, conducts test, performs verification, 

 
\\ navigates back to base. Thinks it is talking to one machine.

It can be expanded to a matched set of calls with no code outside the calls as in…
Test2a()
\\ starts from base/appstate, navigates to test, prepares test up to execution point

Test2b()
\\ sends the execute command minimally, e.g. clicks “OK”.

Test2c()
\\ reacts to test, performs verification, navigates back to base.

The Test2 family still thinks it is talking to one machine. If the test is to see which machine got a lock, for example, the post-execution steps have to be able to handle both ‘getting the lock’ and ‘not getting the lock’ scenarios. The last phase might count how many get the lock. If more than one, an error is reported.   The three test calls in test2 must all take place within the one multitestcase or the automatic reset to base state at the end of a testcase will prevent them from working, so here's an extended version of mMultiTest() that allows you to optionally specify up to three test names to call within the same multitestcase.  Should more be needed, it can be enhanced to as many optional test names as desired without invalidating any existing code (but note the discussion of NULLs in arrays in the following section).

Expanded mMultiTest()

[-]  multitestcase mMultiTest(string aplstate, string testname1, string testname2 optional, string testname3 optional) //triple call multitestcase


[ ] MultiSetup(aplstate)


[ ] string sCurMach


[-] for each sCurMach in lsOnLine



[-] spawn




[ ] SetMachine (sCurMach)




[ ] @(testname1) ()  // this calls the supplied testname by reference


[ ] rendezvous


[-] if (testname2 != NULL)



[ ] string sCurMach2



[-] for each sCurMach2 in lsOnLine




[-] spawn





[ ] SetMachine (sCurMach2)





[ ] @(testname2) ()  // this calls the supplied testname by reference



[ ] rendezvous


[-] if (testname3 != NULL)



[ ] string sCurMach3



[-] for each sCurMach3 in lsOnLine




[-] spawn





[ ] SetMachine (sCurMach3)





[ ] @(testname2) ()  // this calls the supplied testname by reference



[ ] rendezvous


[ ] DisconnectAll()

and our main might look like this:

main()


[ ] mMultiTest(BASE, "test001a", "test001b", "test001c")


[ ] mMultiTest(BASE, "test002a", "test002b")


[ ] mMultiTest(BASE, "test003")

This mMultiTest() function should make it possible to use existing single machine tests with little or no modification. Note that the optional 3rd and 4th arguments in mMultiTest() read a lack of data as a NULL. which is what the if statements are checking for. When no argument is provided, the steps are not executed.

Now we face a trade-off.

In your main() section you can repeatedly call mMultiTest() as above, or you can create an array to feed this information to mMultiTest(). The advantage of the line-at-a-time approach above is that individual tests can be easily commented-out as needed to bypass a broken or unwanted area and goto statements can jump over whole sections. However, using an array for an established series of tests creates simpler and cleaner code. It also gives you a way to group your test calls by functional area, etc. and switch them on or off wholesale. But it is a little more work to maintain if you want to disable a single test as you have to both comment out the test line and change the count of tests. Probably, a mixture of both approaches will be used.

Here’s how to have an array call a list of tests using our extended mMultiTest()

[ ] const BASE = "DefaultBaseState"

[ ] const asFileOpen = "FileOpen", asFileSaveAs = "FileSaveAs" //these are sample app state names

arraytest()


int max = 3



//how many tests in the array?


ARRAY [max][4] OF STRING aTests  
//declare an array list of tests, 4 elements for each test


aTests ={...} 



//populate it



{BASE, "test001a", "test001b", "test001c"}  //list the states and names of tests here



{asFileOpen, "test002a", "test002b",NULL} //Null is required if there is no argument



{asFileOpen, "test003", NULL, NULL}
 


int i


for i = 1 to max



//Don’t spawn this!



mMultiTest (aTests [i,1], aTests [i,2], aTests [i,3], aTests [i,4])
 

 main()


 arraytest()

When arraytest() is called it executes:


test001a() test001b(), and test001c() in order inside one multitestcase


test002a() and test002b() in order inside another multitestcase


test003() inside a third multitestcase

Unfortunately the NULL statements are necessary when populating the array in this relationship. The array structure does not require unused locations to be seeded ({asFileOpen, "test003"} would be valid), and our mMultiTest() function can handle no data in the third and fourth positions because they are optional (mMultiTest (BASE, “test01”) would be valid), but the specific reference to array locations in



mMultiTest (aTests [i,1], aTests [i,2], aTests [i,3], aTests [i,4])
 

will fail if either [i,3] and [i,4] cannot find data in the called array locations, thus a NULL must be used for ‘no data’. If in the future you need to expand mMultiTest to allow more optional tests, you won’t have to expand existing arrays such as the above structure with more nulls unless you expand the call line associated with that array to include aTests[i,5], etc. The above structure would continue to work unmodified.

Adapting existing tests, revisited

By using the above mMultiTest structure, with or without an array, presumably most single-machine based tests can be generalized to multi-machine testing with little or no modification provided: 

1. They are (or can be reduced to) one to three function calls that execute without lines of code in-between

2. They conform to the Segue model of base State-navigate-test-verify-navigate.

3. They use local variables and don’t store data globally.

4. Paths are generalized and compatible to the multi-machine platform

5. A test does not rely on a previous test leaving the application in an intermediate state.

Some testing of elaborate areas is best done by establishing a scenario, then making many tests while in that scenario.  Although this is probably better done with a single machine than with multi-machine testing, it can be done with multi-machine as well. This would require a nest of test functions that are coded specifically for the situation and do not use the mMultiTest() function. It also implies that Verify statements should not be used for intermediate verification.  If you want to test 10 values in one situation, you can create your own version of verify that does not trigger the automatic end of the testcase, and even keep track of passes and fails, writing errors to the results file, then end the testcase with a comparison of the expected number of passes to the actual and use a verify statement at last to confirm the count.

Putting it all together

Now we should have all the parts we need to assemble a master multi-machine testing template. 

1. We need a frame for the application being tested. If two applications are working together, we need two frames and must resolve any conflicts in variable naming for the main window. Often a separate window declaration may be sufficient for the simpler of the two applications.

2. We need any includes that contain basic tools useful in single and multi-machine testing, and lists of test function.

4. We need one master testplan such as mytestplan00.t to initiate and control all the tests, load the includes and hold the main() statement.

5. We may need one or more subordinate testplans (e.g. mytestplan01.t) that contain the new multi-machine tests.

Lead file structure


The “lead” file will be mytestplan00.t This makes local declarations, calls all the includes and calls all the test files.  It has the main statement and this main statement in turn calls all the test files.  To modify test runs, you comment out calls in main() or you use goto statements to skip over sections.

There can be as many variations on mytestplan00.t as needed, each duplicating the includes etc. as needed and having their own specific set of calls under main() This is a simple way to break out a smoke test or a test focused on a specific functional area.

mytestplan00.t


Main test script


Contains:


  Const declarations


  Include calls


  Frame calls


  Test include calls


  Main()


     All test calls

